AM505a — PDE
Solution of Linear Integral Equations

Robert M. Corless

November 25, 1998

References

The main references used for this are [2], [6] and [7]. A. Tadesse used [4, 5, 3] for supplementary material for this morning's lecture.

See http://www.gams.nist.gov for numerical software for the solution of integral equations; the paper [1] is a pointer to the numerical literature.

1 Overview

In this lecture we look at several analytical techniques for solving *linear* one-dimensional integral equations; we look at the Maple share library package IntSolve, written by Honglin Ye [Ph.D. Western]; we look at well-posedness of linear integral equations with smooth kernels; and briefly at approximate and numerical methods. Clearly we're going to have to move at some speed!

You have now seen the general Neumann series, and some examples; you also seen how to solve integral equations with degenerate kernels. This will be quite useful for us when we look at Marchenko's equation for soliton problems.

We also saw briefly this morning an introduction to eigenvalue techniques. We look briefly at some more examples here.

2 Solution Techniques

We now look at some very simple but powerful techniques for some important classes of problems.

2.1 Laplace Transforms

If the integral equation is of *convolution type*, that is, like *Poisson's equation*

$$\phi(x) = \int_0^x K(x-\xi)\phi(\xi)\,d\xi + f(x) \tag{1}$$

or *Abel's equation*

$$\int_0^x K(x-\xi)\phi(\xi)\,d\xi = g(x)\,, \tag{2}$$

then the convolution integrals present suggest that we take Laplace transforms. Remember

$$\mathcal{L}(f)(s) = \int_0^\infty e^{-st} f(t)\,dt \tag{3}$$

$$\mathcal{L}(f*g) = \mathcal{L}(f)\mathcal{L}(g) \tag{4}$$

where the convolution product $f*g$ is

$$\begin{aligned} f*g &= \int_0^x f(x-\xi)g(\xi)\,d\xi \\ &= \int_0^x f(\xi)g(x-\xi)\,d\xi\,. \end{aligned}$$

Hence, Poisson's integral equation becomes

$$\mathcal{L}(\phi) = \frac{\mathcal{L}(f)}{1-\mathcal{L}(K)}$$

provided that $\mathcal{L}(K) \neq 1$. Therefore

$$\phi = \mathcal{L}^{-1}\left(\frac{\mathcal{L}(f)}{1-\mathcal{L}(K)}\right)\,. \tag{5}$$

For example, if our Poisson's equation is

$$\phi(x) = \int_0^x e^{x-y}\phi(y)\,dy + \sin(x)\,, \tag{6}$$

then since $\mathcal{L}(\exp(x)) = 1/(s-1)$ and $\mathcal{L}(\sin(x)) = 1/(s^2+1)$,

$$\begin{aligned}\phi(x) &= \mathcal{L}^{-1}\left(\frac{1}{(s^2+1)(1-\frac{1}{s-1})}\right)\\ &= \mathcal{L}^{-1}\left(\frac{s-1}{(s^2+1)(s-2)}\right)\\ &= \tfrac{1}{5}e^{2x} - \tfrac{1}{5}\cos x + \tfrac{3}{5}\sin x\,. \end{aligned}\tag{7}$$

Substituting this equation back into the original equation shows that this really is a solution.

2.2 Eigenfunction expansions

Suppose we are trying to solve

$$\phi(x) = \int K(x,\xi)\phi(\xi)\,d\xi + f(x)\,, \tag{8}$$

and we know a set of eigenfunctions $\psi_k(x)$ such that

$$\int K(x,\xi)\psi_k(\xi)\,d\xi = \lambda_k\psi_k(x)\,, \tag{9}$$

for appropriate eigenvalues λ_k. We will concern ourselves here only with the case where the set of eigenvalues is finite or at most countably infinite, but in the next two weeks we will see examples of integral equations with a continuous spectrum of eigenvalues. [This is of some importance for physics, and some of you will have seen this already.]

Here, when we expand everything in sight,

$$f(x) = \sum_k b_k\psi_k(x) \tag{10}$$

where the b_k are presumed known because $f(x)$ and the eigenfunctions $\psi_k(x)$ are known, and

$$\phi(x) = \sum_k a_k\psi_k(x) \tag{11}$$

where we want to find the coefficients a_k. Substituting this into the integral equation gives

$$\sum_k (a_k\psi_k - \lambda_k a_k\psi_k - b_k\psi_k) = 0\,.$$

Thus, choosing

$$a_k = \frac{b_k}{1-\lambda_k} \tag{12}$$

solves the problem, unless one of the $\lambda_k = 1$.

For example, consider the Fredholm first kind integral equation (the above analysis was for a second-kind integral equation but it all goes through *mutatis mutandis*):

$$\frac{1}{2\pi}\int_{-\pi}^{\pi}\frac{1-\alpha^2}{1-2\alpha\cos(x-y)+\alpha^2}\,\phi(y)\,dy = f(x)\,. \tag{13}$$

We take $0<\alpha<1$ and $-\pi\le x\le\pi$. The eigenfunctions are just $\cos(kx)$ and $\sin(kx)$. The solution can be shown to be

$$\phi(y) = \tfrac{1}{2}a_0 + \sum_{n\ge1}\alpha^{-n}\left(a_n\cos ny + b_n\sin ny\right)\,, \tag{14}$$

where the a_k and b_k are the Fourier coefficients of f.

One often sees the eigenfunction expansion method applied to degenerate kernels. For example,

$$\phi(y) = \int_0^1\left(1-\frac{x^2y^2}{2}\right)\phi(x)\,dx + \sin\pi y\,. \tag{15}$$

There are only two eigenfunctions. The eigenvalues are $\lambda = 9/20 \pm \sqrt{889}/60$, with associated eigenfunctions $A_i + B_i x^2$. The solution is

$$\phi(y) = \sin\pi y + \tfrac{3}{5}\frac{61\pi^2+20}{\pi^3} - \tfrac{6}{\pi^2}y^2\,. \tag{16}$$

Of course, I did that with Maple, not by hand!

3 Maple code for solution of linear integral equations

In [7] we find a description of `IntSolve`, which resides in the share library of Maple. *Watch out for bugs. The code has "rotted", to use a technical term—that*

is, Maple has evolved since the package was written, and it is not clear that the package still works. In particular I have found some bugs in the eigenfunc method.

4 Solution of Linear Integral Equations in Maple

```
> with(share);
```

```
See ?share and ?share,contents for
information about the share library
```

$$[]$$

```
> with(IntSolve);
```

```
Share Library:  IntSolve

Authors: Ye, Honglin and Corless, Robert.

Description:  Version 1 of
an integral equation solver.
```

$$[\text{“IntSolve”}]$$

```
> ?IntSolve
```

```
> example1 := phi(x) = Int(
exp(x-y)*phi(y), y=0..x) + sin(x);
```

$$example1 := \phi(x) = \int_0^x e^{(x-y)}\,\phi(y)\,dy + \sin(x)$$

```
> p1 := IntSolve( example1, phi(x),
Laplace );
```

$$p1 := \frac{1}{5}\,e^{(2\,x)} - \frac{1}{5}\cos(x) + \frac{3}{5}\sin(x)$$

```
> example2 := phi(y) = Int(
(1-x^2*y^2/2)*phi(x), x=0..1) +
sin(Pi*y);
```

$$example2 := \phi(y) = \int_0^1 (1 - \frac{1}{2}\,x^2\,y^2)\,\phi(x)\,dx + \sin(\pi\,y)$$

```
> p2 := IntSolve( example2, phi(y),
eigenfunc );
```

$$p2 := \sin(\pi\,y) + \frac{3}{5}\,\frac{61\,\pi^2+20}{\pi^3} - 6\,\frac{y^2}{\pi}$$

```
> P2 := unapply(p2,y);
```

$$P2 := y \to \sin(\pi\,y) + \frac{3}{5}\,\frac{61\,\pi^2+20}{\pi^3} - 6\,\frac{y^2}{\pi}$$

```
> value( eval( subs( phi=P2, example2 )
) );
```

$$\sin(\pi\,y) + \frac{3}{5}\,\frac{61\,\pi^2+20}{\pi^3} - 6\,\frac{y^2}{\pi} = -\frac{3}{5}\,\frac{-61\,\pi^2+10\,y^2\,\pi^2-20}{\pi^3} + \sin(\pi\,y)$$

```
> normal( lhs(%) - rhs(%) );
```

$$0$$

```
> example3 := phi(x) = lambda*Int(
exp(k*(x-y))*phi(y), y=0..1) + sin(x);
```

$$example3 := \phi(x) = \lambda \int_0^1 e^{(k\,(x-y))}\,\phi(y)\,dy + \sin(x)$$

Two terms of the Neumann series:

```
> IntSolve( example3, phi(x), 2 );
```

$$(\sin(x)\,k^2 + \sin(x) - \lambda\,e^{(k\,(x-1))}\,k\sin(1) - e^{(k\,(x-1))}\,\lambda^2\,k\sin(1) - e^{(k\,(x-1))}\,\lambda^2\cos(1) - \lambda\,e^{(k\,(x-1))}\cos(1) + \lambda^2\,e^{(k\,x)} + \lambda\,e^{(k\,x)})/(k^2+1)$$

Can try to convert to a differential equation:

```
> p3 := IntSolve( example3, phi(x),
differentiate);
```

$$p3 := \left\{ \mathrm{D}(\phi)(x) - k\,\phi(x) - \cos(x) + k\sin(x) = 0,\ \phi(0) - \lambda \int_0^1 e^{(-k\,y)}\,\phi(y)\,dy = 0 \right\}$$

```
> P3sol := dsolve(p3[1], phi(x));
```

$$P3sol := \phi(x) = \sin(x) + e^{(k\,x)}\,\_C1$$

```
> P3 := unapply( rhs(P3sol), x );
```

$$P3 := x \to \sin(x) + e^{(k\,x)}\,\_C1$$

```
> eval(subs(phi=P3,p3[2]));
```

$$\_C1 - \lambda(-\cos(1)\,e^{(-k)} - k\,e^{(-k)}\sin(1) + \_C1\,k^2 + \_C1 + 1)/(k^2+1) = 0$$

```
> map(normal, readlib(isolate)(%,_C1)
);
```

$$\_C1 = \frac{\lambda\,(\cos(1)\,e^{(-k)} + k\,e^{(-k)}\sin(1) - 1)}{-k^2 - 1 + \lambda\,k^2 + \lambda}$$

That's very impressive, but is it right?

```
> assign(%);
> value( eval( subs(phi=P3, example3 )
) );
```

$$\sin(x) + \frac{e^{(k\,x)}\,\lambda\,(\cos(1)\,e^{(-k)} + k\,e^{(-k)}\sin(1) - 1)}{-k^2 - 1 + \lambda\,k^2 + \lambda} =$$
$$\lambda(k\,e^{(k\,x-k)}\sin(1) - \lambda\,k\sin(1)\,e^{(k\,x-k)} + \lambda\,k\,e^{(-k)}\sin(1)\,e^{(k\,x)} + \lambda\cos(1)\,e^{(-k)}\,e^{(k\,x)} + \cos(1)\,e^{(k\,x-k)} - \lambda\cos(1)\,e^{(k\,x-k)} - e^{(k\,x)})/((k^2+1)\,(-1+\lambda)) + \sin(x)$$

```
> simplify( normal( lhs(%) - rhs(%) )
);
```

$$0$$

5 Well-posedness

There are many methods to approximate the solution of an integral equation. There are many methods to solve them numerically. But before we show how to solve problems approximately, we observe that linear integral equations with smooth kernels are well-posed. Suppose for example we have somehow computed an approximate solution $\theta(x)$ to the Fredholm 2nd kind equation (8). Define the *residual*

$$r(x) := \theta(x) - \int K(x,\xi)\theta(\xi)\,d\xi - f(x)\,. \qquad (17)$$

Then we have found the *exact solution* to the related integral equation

$$\psi(x) = \int K(x,\xi)\psi(\xi)\,d\xi + \hat{f}(x)\,, \qquad (18)$$

where $\hat{f}(x) = f(x) + r(x)$. This is trivial. But it's also profound. If $r(x)$ is small *compared to the approximations already made in deriving the integral equation, or to physically reasonable perturbations in f, then we are done.*

Perhaps instead we have exactly computed a solution satisfying an equation with a modified kernel (one technique is to replace $K(x,\xi)$ with a degenerate kernel $\hat{K}$ that approximates K, for example). Is $\hat{K} - K$ smaller than terms neglected in the model? Yes? *Done.*

This raises the question of how sensitive the solution of the integral equation is to changes in f or K. In general this is a deep subject (with singular kernels, for example). But for our case, we find in [2, p. 156] that there are computable constants N^* and C such that

$$|\hat{\phi} - \phi| \le \frac{(\varepsilon N^* + \eta)(1+C)}{1 - \varepsilon(1+C)} \qquad (19)$$

where ε is a bound on the distance $||\hat{K} - K||$ and η is a bound on $||\hat{f} - f||$. This is as good a behaviour as can be expected, and shows that not only is the problem well posed, we have a linear dependence on the perturbations.

6 Collocation

This is a reasonable numerical method to solve many kinds of integral equations, not just the simple ones here. See however the many numerical papers, starting perhaps from those of Hermann Brunner (Memorial U.), e.g. [1], or the software available through http://www.gams.nist.gov, for real methods. The following is just a sketch.

Choose points ("collocation points") x_k on $a \le x \le b$. Then

$$\phi(x) = \int_a^b K(x,\xi)\phi(\xi)\,d\xi + f(x)$$

implies

$$\phi(x_k) = \int_a^b K(x_k, \xi)\phi(\xi)\, d\xi + f(x_k)\,. \tag{20}$$

Now if we approximate $\phi(y)$ by a linear combination of some chosen "gauge functions" $\psi_j(y)$ (say B-splines, or piecewise constants, or whatever), namely

$$\phi(y) = \sum_j \alpha_j \psi_j(y)\,, \tag{21}$$

then substitution into (20) gives a set of *linear* equations for the unknown coefficients α_j, which depend on the integrals $\int_a^b K(x,\xi)\psi_j(\xi)\, d\xi$. We choose the guage functions to make these simple to evaluate, of course.

6.1 Example

Suppose we wish to solve

$$\phi(x) = \int_0^1 e^{xy}\phi(y)\, dy + \frac{1}{1+x^2} \tag{22}$$

Choose $x_k = (k-1/2)/N$ for $1 \le k \le N$, equally spaced on $0 \le x \le 1$. Choose our gauge functions to be piecewise constant, $\psi_j(x) = \psi_j$ on $(k-1)/N \le x \le k/N$ and zero otherwise. Then our collocation equations become

$$\psi_j = \sum_{k=1}^{N} \psi_k \int_{(k-1)/N}^{k/N} e^{y(j-1/2)/N}\, dy + \frac{1}{1+\left(\frac{j-1/2}{N}\right)^2}\,, \tag{23}$$

and rearranging this gives a simple linear system

$$(A-I)\vec{\psi} = \left\{ \frac{-1}{1+\left(\frac{j-1/2}{N}\right)^2} \right\}$$

where the matrix entries are $A_{ij} =$

$$\frac{N}{i-1/2}\left[\exp\left(\frac{(i-1/2)j}{N^2}\right) - \exp\left(\frac{(i-1/2)(j-1)}{N^2}\right)\right]. \tag{24}$$

We hope that as $N \to \infty$, the $\psi_j \to \phi(x_j)$. As the following Maple session shows, it works.

```
> restart;
```

```
> phi(x) = Int( exp(x*y)*phi(y),
y=0..1) + 1/(1+x^2);
```

$$\phi(x) = \int_0^1 e^{(x\,y)}\,\phi(y)\, dy + \frac{1}{1+x^2}$$

```
> um := %:
> IntSolve( um, phi(x),2 );
Warning, computation interrupted
> restart;
```

```
> p := proc(N::posint) local A, b;
>   A := matrix(N,N,
>   (i,j)->evalf((exp((i-1/2)*j/N^2)
>     -exp((i-1/2)*(j-1)/N^2))*N/(i-1/2)));
>   b := vector(N, i->evalf(1/(1+(i-1/2)^2/N^2)) );
>   linalg[linsolve](evalm(-A+&*()),b)
> end:
> p(2);
```

$$[-1.595537339, -2.795225413]$$

```
> p(4);
```

$$[-1.248020298, -1.721998520, -2.319643386, -3.002682256]$$

```
> p(10);
```

$$[-1.103197136, -1.253032875, -1.430046122, -1.631232552, -1.852947221, -2.091770186, -2.345087651, -2.611336610, -2.889992145, -3.181416068]$$

```
> p(30);
```

$$[-1.053813402, -1.097363590, -1.144113251, -1.194031617, -1.247062256, -1.303125609, -1.362122403, -1.423937411, -1.488443464, -1.555505672, -1.624985325, -1.696743735, -1.770645567, -1.846561767, -1.924371993, -2.003966513, -2.085247612,$$

$-2.168130508, -2.252543882,$
$-2.338430027, -2.425744663,$
$-2.514456593, -2.604547123,$
$-2.696009313, -2.788847278,$
$-2.883075290, -2.978716985,$
$-3.075804529, -3.174377822,$
$-3.274483757]$

```
>  p30 := %:
```

```
>  plot([seq( [(k-1/2)/30,
p30[k]],k=1..30 )] );
```




```
>  ph := unapply( piecewise( seq(
op([x<k/30,p30[k]]), k=1..30 )), x);
```

$$ph := x \to \text{piecewise}(x < \frac{1}{30}, -1.053813402, x < \frac{1}{15},$$
$$-1.097363590, x < \frac{1}{10}, -1.144113251,$$
$$x < \frac{2}{15}, -1.194031617, x < \frac{1}{6}, -1.247062256,$$
$$x < \frac{1}{5}, -1.303125609, x < \frac{7}{30}, -1.362122403,$$
$$x < \frac{4}{15}, -1.423937411, x < \frac{3}{10},$$
$$-1.488443464, x < \frac{1}{3}, -1.555505672, x < \frac{11}{30},$$
$$-1.624985325, x < \frac{2}{5}, -1.696743735, x < \frac{13}{30},$$
$$-1.770645567, x < \frac{7}{15}, -1.846561767, x < \frac{1}{2},$$
$$-1.924371993, x < \frac{8}{15}, -2.003966513,$$
$$x < \frac{17}{30}, -2.085247612, x < \frac{3}{5}, -2.168130508,$$
$$x < \frac{19}{30}, -2.252543882, x < \frac{2}{3}, -2.338430027,$$
$$x < \frac{7}{10}, -2.425744663, x < \frac{11}{15},$$
$$-2.514456593, x < \frac{23}{30}, -2.604547123, x < \frac{4}{5},$$
$$-2.696009313, x < \frac{5}{6}, -2.788847278, x < \frac{13}{15},$$
$$-2.883075290, x < \frac{9}{10}, -2.978716985,$$
$$x < \frac{14}{15}, -3.075804529, x < \frac{29}{30},$$
$$-3.174377822, x < 1, -3.274483757)$$

```
>  term := unapply( int(
exp(x*y),y=(k-1)/30..k/30), x,k);
```

$$term := (x, k) \to \frac{e^{(1/30\,x\,k)} - e^{(1/30\,x\,(k-1))}}{x}$$

```
>  res := x -> evalf( ph(x) - add(
p30[k]*term(x,k), k=1..30) - 1/(1+x^2)):
```

```
>  res(0.1);
```

$$-.0253523379$$

```
>  plot(res, 0..1, -0.05..0.05);
```



References

[1] J. Blom and H.Brunner, ACM TOMS 17 (1991) pp. 167-177.

[2] Ll. G. Chambers, *Integral Equations: A Short Course*, International Textbook Company Ltd., 1976.

[3] M. A. McKiernan and J. Wainwright, *Advanced analytical methods in applied mathematics*, lecture notes for AM964a, 1981.

[4] A. C. Pipkin, *A course on integral equations*, Texts in Applied Mathematics, Springer-Verlag, 1991.

[5] W. Pogorzelski, *Integral equations and their applications*, vol. 1, Pergamon, Oxford, 1966.

[6] S. L. Sobolev, *Partial Differential Equations of Mathematical Physics*, Dover, 1964.

[7] Honglin Ye and Robert M. Corless, "Solving linear integral equations with Maple", Proc. ISSAC Berkeley, CA, July 27–29 (1992) 95–103